S1. OxCal Code

Model 1: without reservoir


Plot()
 {
  Sequence("main")
  {
   Boundary("start");
   Phase("Below and early channels")
   {
    Sequence("UbaidBelowCH3")
    {
     Boundary("startUbaid4",Date(BC(4900)));
     Phase("Ubaid")
     {
      R_Date("13012_TG38",6020,30);
      R_Date("16006",5750,30);
      R_Date("13007_TG33",5710,30);
      R_Date("13009_TG35",5640,30);
     };
     Boundary("endUbaid4",Date(BC(4500)));
     R_Date("TG-E6",5590,30);
     Boundary("PostUbaid4");
    };
    Sequence("CH3-early")
    {
     Boundary("CH3-earlystart");
     R_Date("TH2-SH1-145",6860,30);
     R_Date("TH12-SH1-40",6090,30);
     Boundary("CH3-earlyend");
    };
    Sequence("CH2sequence")
    {
     Boundary("startCH2phase1");
     R_Date("TH10-SH1-80",6050,30);
     R_Date("TH1-SH4-40",5930,30);
     Boundary("endCH2phase1");
    };
    Sequence("CH7sequence")
    {
     Boundary("startCH7phase1");
     R_Date("TH3-SH1-400",6520,30);
     R_Date("TH8-SH2-75",6280,30);
     R_Date("TH8-SH1-60",4940,30);
     Boundary("endCH7phase1");
    };
   };
   Boundary("EDIstart",Date(BC(2900)));
   Phase("Channels_Finalphase")
   {
    Phase(CH6E)
    {
     R_Date("TH9-CH1-70",4080,30);
    };
    Sequence("HarbourSequence")
    {
     Boundary("startHarbour");
     R_Date("TH13-SH6-130",4380,30);
     R_Date("TH13-SH1-76",3850,30);
     Boundary("endHarbour");
    };
    Sequence("CH5sequence")
    {
     Boundary("startCH5");
     R_Date("TH6-SH5-120",4440,30);
     R_Date("TH6-SH4-100",6210,30)
     {
      Outlier();
     };
     Boundary("endCH5");
    };
    Sequence("CH6Wsequence")
    {
     Boundary("startCH6W");
     R_Date("TH7-SH3-100",4370,30);
     R_Date("TH7-SH2-90",4130,30);
     Boundary("endCH6W");
    };
    Sequence("CH2-1-3")
    {
     Boundary("startCH2late");
     R_Date("TH15-SH2-35",4410,30);
     R_Date("TH14-SH1-60",5840,30)
     {
      Outlier();
     };
     R_Date("16014-SH2-320",4150,30);
     Boundary("endCH3late");
    };
   };
   Boundary("OldBabylonian",Date(BC(1800)));
  };
 };
















Model 2: with reservoir correction


Plot()
 {
  Sequence("main")
  {
   Curve("IntCal20","intcal20.14c");
   Curve("Offset","intcal20.14c")
   {
    Reservoir(345,60);
   };
   Boundary("start");
   Phase("Below and early channels")
   {
    Sequence("UbaidBelowCH3")
    {
     Boundary("startUbaid4",Date(BC(4900)));
     Phase("Ubaid")
     {
      Curve("=Offset");
      R_Date("13012_TG38",6020,30);
      Curve("=IntCal20");
      R_Date("16006",5750,30);
      R_Date("13007_TG33",5710,30);
      R_Date("13009_TG35",5640,30);
     };
     Boundary("endUbaid4",Date(BC(4500)));
     Curve("=IntCal20");
     R_Date("TG-E6",5590,30);
     Boundary("PostUbaid4");
    };
    Sequence("CH3-early")
    {
     Boundary("CH3-earlystart");
     Curve("=Offset");
     R_Date("TH2-SH1-145",6860,30);
     R_Date("TH12-SH1-40",6090,30);
     Boundary("CH3-earlyend");
    };
    Sequence("CH2sequence")
    {
     Boundary("startCH2phase1");
     Curve("=Offset");
     R_Date("TH10-SH1-80",6050,30);
     R_Date("TH1-SH4-40",5930,30);
     Boundary("endCH2phase1");
    };
    Sequence("CH7sequence")
    {
     Boundary("startCH7phase1");
     Curve("=Offset");
     R_Date("TH3-SH1-400",6520,30);
     R_Date("TH8-SH2-75",6280,30);
     R_Date("TH8-SH1-60",4940,30);
     Boundary("endCH7phase1");
    };
   };
   Boundary("EDIstart",Date(BC(2900)));
   Phase("Channels_Finalphase")
   {
    Phase(CH6E)
    {
     Curve("=IntCal20");
     R_Date("TH9-CH1-70",4080,30);
    };
    Sequence("HarbourSequence")
    {
     Boundary("startHarbour");
     Curve("=Offset");
     R_Date("TH13-SH6-130",4380,30);
     R_Date("TH13-SH1-76",3850,30);
     Boundary("endHarbour");
    };
    Sequence("CH5sequence")
    {
     Boundary("startCH5");
     Curve("=Offset");
     R_Date("TH6-SH5-120",4440,30);
     R_Date("TH6-SH4-100",6210,30)
     {
      Outlier();
     };
     Boundary("endCH5");
    };
    Sequence("CH6Wsequence")
    {
     Boundary("startCH6W");
     Curve("=Offset");
     R_Date("TH7-SH3-100",4370,30);
     R_Date("TH7-SH2-90",4130,30);
     Boundary("endCH6W");
    };
    Sequence("CH2-1-3")
    {
     Boundary("startCH2late");
     Curve("=Offset");
     R_Date("TH15-SH2-35",4410,30);
     R_Date("TH14-SH1-60",5840,30)
     {
      Outlier();
     };
     R_Date("16014-SH2-320",4150,30);
     Boundary("endCH3late");
    };
   };
   Boundary("OldBabylonian",Date(BC(1800)));
  };
 };
